
Gráficos em Visual Basic
Introdução

Os gráficos podem ser vetoriais ou mapas de bits. Gráficos vetoriais são gerados por comandos e equações sendo independentes da resolução da exibição não necessitando ser armazenados em forma gráfica. Mapas de bits são imagens processadas pixel a pixel e são dependentes da resolução não podendo ser ampliados sem perda de resolução, além de necessitar de armazenamento em forma gráfica.
Em VB os gráficos podem ser colocados nos componentes formulário, PictureBox e ImageBox. O componente ImageBox só aceita mapas de bits enquanto os demais componentes aceitam gráficos vetoriais e mapas de bits.

Mapas de bits são carregados para a memória enquanto os gráficos vetoriais são gerados no momento da exibição. Pra carregar um gráfico em um componente em tempo de projeto se atribui o nome do arquivo aonde está armazenado o gráfico à propriedade Picture do componente, na janela d propriedades. Para atribuir ou atualizar uma imagem em tempo de execução usa-se a função LoadPicture().

Dimensionamento de Imagens

O componente PictureBox dispõe de uma propriedade AutoSize. Quando esta propriedade for igual a True o componente é redimensionado para se acomodar às dimensões da imagem. Em caso contrário exibe-se apenas a porção da imagem que caiba no tamanho corrente do componente. O componente ImageBox dispõe d uma propriedade Stretch. Quando esta propriedade for igual a True a imagem é redimensionada para se ajustar às dimensões correntes do componente, mesmo com eventuais distorções da imagem.

Carga e gravação de Imagens

Em tempo de execução a carga de imagens é feita da forma

<Componente>.Picture = LoadPicture(<nome do arquivo>)
Caso a imagem tenha sido alterada durante o processamento e se desejar gravar a nova imagem se escreve

SavePicture <figura>, <nome de arquivo>

Uso da Área de Transferencia

Os mapas de bits armazenados na área de transferencia são referenciados por uma constante simbólica denominada vbCFBitmap. As cópias para e da área de transferencia são feitas da forma

Clipboard.SetData Figura1.Image, vbCFBitmap

Figura2.Image = Clipboard.GetData()

Métodos de Desenho

Os métodos de desenho em VB são os seguintes:

Método
Descrição

Print
Exibe um string

Line
Desenha segmentos de reta e caixas

Circle
Desenha circunferências e arcos

Point
Retorna o valor da cor em um ponto

PSet
Atribui uma cor a um ponto

Para desenhar segmentos de reta a sintaxe é:

Line [Step] (X1,Y1) - [Step] (X2,Y2), [cor], [B], [F]

aonde:


Descrição

(X1,Y1)
Coordenadas do ponto inicial do segmento

(X2,Y2)
Coordenadas do ponto final do segmento

Step
Indicador do emprego de coordenadas relativas

cor
Especificação da cor do segmento

B
Especificação que se deseja desenhar uma caixa (box)

F
Especificação que se deseja dar um enchimento a uma forma fechada

A cor default do segmento será aquela especificada em ForeColor do componente. Caso seja desejada uma cor diferente para o segmento a desenhar a opção cor pode receber qualquer expressão válida, como, por exemplo RGB(255,0,0) ou &H0000FF.

Quando se utiliza coordenadas relativas a origem sempre é o ponto corrente. O comando

Line (100,100) - (300,200)

traça um segmento do ponto de coordenadas (100,100) até o ponto de coordenadas (300,200). O comando

Line (100,100) - Step (300,200)

traça um segmento do ponto de coordenadas (100,100) até o ponto de coordenadas (400,300).

Para construir um retângulo com vértices nos pontos (100,100), (400,100), (400,200) e (100,200) pode-se usar qualquer dos seguintes comandos:

Em coordenadas absolutas

Line (100,100) - (400,100)

Line (400,100) - (400,200)
Line (400,200) - (100,200)
Line (100,2100) - (100,100)
Em coordenadas relativas

Line (100,100) - Step (300,0)

Line - Step (0,100)

Line - Step (-300,0)

Line - Step (0,-100)

Uma alternativa é empregar a opção B. Caso esta opção esteja incluída basta fornecer o vértice superior esquerdo e o vértice inferior direito. Assim, as duas maneiras de desenhar caixas anteriores podem ser substituídas por:

Line (100,100) - (400,200), , B

Line (100,100) - Step (300,100), ,B

O preenchimento de formas fechadas pode ser feito com a opção F que pode aceitar valores para o estilo do enchimento e também para a cor de enchimento.

Para desenhar circunferências a sintaxe é:

Circle [Step] (X,Y), <raio> , [cor], [ponto inicial], [ponto final]. [aspecto]

aonde:


Descrição

Step
Indicador do emprego de coordenadas relativas

(X,Y)
Coordenadas do centro da circunferência

Raio
Raio da circunferência

cor
Especificação da cor da circunferência

Ponto inicial
Ponto inicial de um arco, medido em radianos

Ponto final
Ponto final de um arco, medido em radianos

aspecto


Quando os pontos inicias e finais forem precedidos do sinal ‘-‘ isto significa que o ponto correspondente deve ser conectado ao centro da circunferência, como nos gráficos tipo torta.

Desenho de Curvas

Para desenhar uma curva cujos limites ao longo do eixo dos X sejam Xmin e Xmax, sendo o número de pixels ao longo do eixo dos X igual Xpixels, calcula-se a coordenada corrente ao longo do eixo dos X, para caracterizar cada pixel, da forma

For i = 0 To XPixels - 1

    t = XMin + (XMax - XMin) * i / XPixels

Next

Caso se desejasse calcular y = sen(2t)e2/t então a oordenada seria calculada da forma

Function FunctionEval1(ByVal X As Double) As Double

    FunctionEval1 = Exp(2 / X) * Cos(2 * X)

End Function

A repetição principal, para a determinação dos pontos da curva, seria

For i = 0 To XPixels

    t = XMin + (XMax - XMin) * i / XPixels

    'Picture1.PSet (t, FunctionEval2(t))

    Picture1.Line -(t, FunctionEval1(t))

Next

A obtenção de XPixels é feita da seguinte maneira

Picture1.ScaleMode = 3

XPixels = Picture1.ScaleWidth - 1

Depois de calculadas as ordenadas extremas da curva, Ymax e Ymin, reajusta-se o sistema de coordenadas para as coordenadas do usuário

Picture1.Scale (XMin, YMin)-(XMax, YMax)

A marcação dos pontos do gráfico é feita por Figura1.Pset(t,FunctionEval1(t)). Para tornar o desenho mais continuo pode-se substituir a marcação de pontos pelo traçado de segmentos entre os pontos vizinhos, da forma

Figura1.PSet (t, FunctionEval1(t))

O código completo para o traçado de duas curvas ( y = sen(2t)e2/t  e y = cos(3t)sen(5t))é mostrado a seguir.

Function FunctionEval1(ByVal X As Double) As Double

    FunctionEval1 = Exp(2 / X) * Cos(2 * X)

End Function

Function FunctionEval2(ByVal X As Double) As Double

    FunctionEval2 = Cos(3 * X) * Sin(5 * X)

End Function

Private Sub Command1_Click()

Dim t As Double

Dim XMin, XMax, YMin, YMax As Double

YMin = 1E+101

YMax = -1E+101

XMin = 2

XMax = 10

Picture1.Cls

Picture1.ScaleMode = 3

XPixels = Picture1.ScaleWidth - 1

' Cálculo dos valores extremos para o eixo dos Y 

For i = 1 To XPixels

    t = XMin + (XMax - XMin) * i / XPixels

    functionVal = FunctionEval1(t)

    If functionVal > YMax Then YMax = functionVal

    If functionVal < YMin Then YMin = functionVal

Next

' Estabelecimento de um sistema de coordenadas definido pelo usuário

Picture1.Scale (XMin, YMin)-(XMax, YMax)

Picture1.ForeColor = RGB(0, 0, 255)

' Movimento para o ponto inicial

Picture1.PSet (XMin, FunctionEval1(XMin))

' Traçado da função

For i = 0 To XPixels

    t = XMin + (XMax - XMin) * i / XPixels

    'Picture1.PSet (t, FunctionEval2(t))

    Picture1.Line -(t, FunctionEval1(t))

Next

End Sub

Private Sub Command2_Click()

Dim t As Double

Dim XMin, XMax, YMin, YMax As Double

YMin = 1E+101

YMax = -1E+101

XMin = 2

XMax = 10

Picture1.Cls

Picture1.ScaleMode = 3

XPixels = Picture1.ScaleWidth - 1

' Cálculo dos valores extremos para o eixo dos Y 

For i = 0 To XPixels

    t = XMin + (XMax - XMin) * i / XPixels

    functionVal = Cos(3 * t) * Sin(5 * t)

    If functionVal > YMax Then YMax = functionVal

    If functionVal < YMin Then YMin = functionVal

Next

' Estabelecimento de um sistema de coordenadas definido pelo usuário

Picture1.Scale (XMin, YMin)-(XMax, YMax)

Picture1.ForeColor = RGB(255, 0, 0)

' Traçado da função

For i = 0 To XPixels - 1

    t = XMin + (XMax - XMin) * i / XPixels

    functionVal = Cos(3 * t) * Sin(5 * t)

    'Picture1.PSet (t, functionVal)

    Picture1.Line -(t, functionVal)

Next

End Sub

Private Sub Command3_Click()

Dim t As Double

Dim XMin, XMax, YMin, YMax As Double

YMin = 1E+101

YMax = -1E+101

XMin = 2

XMax = 10

Picture1.Cls

Picture1.ScaleMode = 3

XPixels = Picture1.ScaleWidth - 1

' Cálculo dos valores extremos para o eixo dos Y 

For i = 1 To XPixels

    t = XMin + (XMax - XMin) * i / XPixels

    functionVal = FunctionEval1(t)

    If functionVal > YMax Then YMax = functionVal

    If functionVal < YMin Then YMin = functionVal

Next

' Estabelecimento de um sistema de coordenadas definido pelo usuário

Picture1.Scale (XMin, YMin)-(XMax, YMax)

Picture1.ForeColor = RGB(0, 0, 255)

' Movimento para o ponto inicial

Picture1.PSet (XMin, FunctionEval1(XMin))

' Traçado da função

For i = 0 To XPixels

    t = XMin + (XMax - XMin) * i / XPixels

    'Picture1.PSet (t, FunctionEval2(t))

    Picture1.Line -(t, FunctionEval1(t))

Next

Picture1.ForeColor = RGB(255, 0, 0)

Picture1.PSet (XMin, FunctionEval2(XMin))

' Traçado da função

For i = 0 To XPixels

    t = XMin + (XMax - XMin) * i / XPixels

    'Picture1.PSet (t, FunctionEval2(t))

    Picture1.Line -(t, FunctionEval2(t))

Next

End Sub

